
要旨 

これまでSASの初心者が戸惑うことの一つとして、SASプログラムでは自作の関数とサブルーティ

ンを作成する機能がないことが挙げられる。筆者は、最近まで LINK-RETURN文やマクロ言語などを

使うことで対処してきたが、数年前(SAS9.2 以降)から PROC FCMP を使って関数とサブルーティンの

作成が可能となり、大変便利になった。本論文では、同じ処理をする SAS プログラムを、①初心者

レベル、②ARRAY文使用、③LINK-RETURN文使用、④マクロ言語使用、⑤PROC FCMPを使って関数

作成、⑥PROC FCMPを使ってサブルーティン作成の、６通りで作成して解説する。 

 

キーワード PROC FCMP､関数､サブルーティン､LINK-RETURN文､マクロ言語､ARRAY文 

1. はじめに 

筆者は SAS を使い始めて 30 数年になる。それまでは、FORTRAN や Algol でプログラミ

ングをしてきた者にとって、SAS を使い始めた頃は戸惑うことが多かった。当時は今のよ

うに、分からないことはネット検索をすれば大抵のことが分かる時代ではなく、SAS のマ

ニュアルも BASE と STAT の薄い 2 冊があるだけで、周りに SAS をよく知っている人もほ

とんどいない状況だった。そういう手探りの中で SAS のプログラミングを始めたのだが、

当然困ることがいくつも出てきた。その一つが、SAS プログラムでは自作の関数とサブルーテ

ィンを作成する機能がないことであった。だんだん SAS を使っていくうちに、LINK-RETURN 文、

ARRAY文、マクロ言語を憶え、それらを使って関数やサブルーティンの代用品として対処してきたが、

数年前(SAS9.2以降)から PROC FCMPを使って関数とサブルーティンの作成が可能となり、大変便利

になった。本論文では、簡単な例題を取り上げて、SAS を使い始めたユーザーにも分かるように、

同じ処理をするSASプログラムを、①初心者レベル、②ARRAY文使用、③LINK-RETURN文使用、④

マクロ言語使用、⑤PROC FCMP を使って関数作成、⑥PROC FCMP を使ってサブルーティン作成の、

６通りで作成して解説する。 

2. 例題 
表１に示す架空の 5 名の個人情報のエクセルファイルを取り上げる。

同表には各自の①ID 番号、②生年、③結婚した年の三つの情報が

含まれている。②や③には同表に示すように、西暦と和暦が混在して

いる。和暦は、例えば、平成３年なら「平 3」、明治 44 年なら「明 44」の

ように表記されている。③から②を引き算して、結婚した時の年齢を計

算する SAS プログラムを作成する。引き算をするためには、和暦を全

て西暦に換算しなければならない。最終的には、その換算をする関数

表１ 婚姻年齢.xlsx 

① ② ③

idno birthyr wedyr

A1001 昭25 1983

A1002 平3 平26

A1003 大14 昭30

A1004 1989 2014

A1005 明44 昭21



とサブルーティンを作成するのが目的である。次節以降で６種類の処理方法を解説するが、プログ

ラムそのものは、便宜上、一つの SAS ファイル(婚姻年齢.sas)に納めている（付録参照）。 
 

3. エクセル表を SAS データセットに変換 
右のプログラムを実行すると、表 1のエクセル表が図１に示すパーマネント SAS データセット

wedageに変換される。 

 

 

 

 

 

 

 

4. 和暦を西暦に換算するアルゴリズム 
表 1 の和暦表現は明治、大正、昭和、平成の先頭の漢字一文字とそれに続く１～２桁の半角数

字から成るので、文字変数としては、長さ４バイトになる。和暦か西暦かの区別は、先頭の２バイトが

「明大昭平」か「19」か「20」のいずれかであるかによって判定できる。SELECT-WHEN 文を使って

和暦ならそれぞれの「元年の西暦年－１」を変数 startyearに設定しておき、後で３～４バイト目の

数字をstartyearに加算すれば、西暦に換算できる。また、初めから西暦ならstartyearに一旦０

を設定しておき、後で元の西暦年の値を加算すればよい。プログラムでは、欠損値または表記エラ

ーのケースも想定し、othrewise として startyearに数値型の欠損値「半角ピリオド」を設定してお

く。ちなみに、生年の換算だけを取り出すと、このようなプログラムになる。 

 

 

 

 

 

 

5. 初心者レベルのプログラム 

前節で生年の西暦換算のアルゴリズムを示したが、結婚した年でも同じ換算処理が必要になる。

取りあえず、そのアルゴリズムをコピー・ペーストした後、変更が必要な箇所を修正して出来たのが

以下に示すプログラムで、実行結果も示す(図２)。これはこれで今は問題ないが、仮に、両親の生

年もデータに含まれていて、本人が生まれた年の両親の年齢も計算する場合を考えると、コピー・

ペースト作業が更に 2 回必要になり、当然その都度変数名等を修正することになる。こういうコピ

ー・ペースト作業を繰り返していると、うっかり変数名の修正をやり損なう箇所が出てきてもおかしく

ないし、プログラムを書いた経験のある人なら､そのようなミスをしたことは必ずあると思う。とりあえず

/* 婚姻年齢.sas */ options nocenter; 
libname FCMP "G:\sasFCMP\sasds"; 
 
proc import out=FCMP.wedage 
    datafile="G:\sasFCMP\婚姻年齢.xlsx" 
    dbms=excel replace; 
    getnames=yes; 
    sheet="sheet1"; 
run; 
 
proc print; title "data=wedage"; run; 

select(substr(birthyr,1,2)); *★★生年を西暦変換★★; 
when("明") startyear=1867; 
when("大") startyear=1911; 
when("昭") startyear=1925; 
when("平") startyear=1988; 
when("19") startyear=0; 
when("20") startyear=0; 
otherwise  startyear=.; *欠損値又はエラー; 

end; 
if startyear > 0 then birthyr=startyear+substr(birthyr,3,2); 

                   else birthyr=startyear+birthyr; 

図１ データセット wedage 



正常に作動すれさえすれば良いと思ってプログラムを書いているユーザーにこのようなプログラム

が多いが、高度なプログラムを作成するには限界がある。 

6. ARRAY 文使用のプログラム 
同じアルゴリズムのコピー・ペーストの回避策として、array 文を使う方法がある。つまり、同じア

ルゴリズムで処理をする対象の全ての変数をひとつの arrayの中に並べてから、doループで回せ

ばよい。具体的には、

「array arg $ birthyr 

wedyr; 」 と すれば 、

arg{1}と arg{2}はそ

れぞれ変数 birthyr と 

wedyr に対応している

ので、do ループから

出てきた時には、両

方の変数とも西暦換

算済みにある。 

（六通りのプログラムの実行
結果は、title 以外は図 2 と

全く同じなので省略する。） 

 

*★★★ ①初心者のプログラム ★★★; 
data beginner (drop=startyear); set FCMP.wedage; 
  select(substr(birthyr,1,2)); *★★生年を西暦変換★★; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then birthyr=startyear+substr(birthyr,3,2); 
                 else birthyr=startyear+birthyr; 
 
  select(substr(wedyr,1,2)); *★★結婚した年を西暦変換★★; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then wedyr=startyear+substr(wedyr,3,2); 
                   else wedyr=startyear+wedyr; 
 
  wedage=wedyr-birthyr; *結婚した時の年齢計算; 
run; 
 
proc print data=beginner; title "①初心者のプログラム"; run; 

図２ 実行結果 

*★★★ ②ARRAY 文使用のプログラム ★★★; 
data array_used (drop=i startyear); set FCMP.wedage; 
  array arg $ birthyr wedyr; 
  do i=1 to 2; 
    select(substr(arg{i},1,2)); *★★年を西暦変換★★; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then arg{i}=startyear+substr(arg{i},3,2); 
                     else arg{i}=startyear+arg{i}; 
  end; 
  wedage=wedyr-birthyr; *結婚した時の年齢計算; 
run; 
 
proc print data=array_used; title "②ARRAY 文使用"; run; 



7. LINK-RETURN 文使用のプログラム 
SAS にはこれまで自作の関数やサブルーティンを定義して利用する機能がなかったが、筆者は

この対策の一つとして使ってきたのが LINK-RETURN文である。以下に示す data ステップの先頭部

がいわゆるメインプログラムに相当し、それに続く箇所が「開いたサブルーティン」に相当する。 

LINK-RETURN文の仕組みは以下の①～④で解説している。 
 

① 実引数に相当する変数 birthyrを仮の引数に相当する変数 YEARに代入する。この時の

「引数」の個数には制限はない。 

② LINK 文で「開いたサブルーティン」の先頭行の行ラベルを指定する。ここで指定した行ラ

ベル CONVERTから RETURN文までが開いたサブルーティンに相当する。 

③ この LINK文が実行されると、そこで指定されたラベル行にプログラムの制御が移る。GOTO

文と同じように見えるが、ここでは GOTO 文は使えない。理由は、GOTO 文では後で自動的に

元の箇所に制御を戻すことができないからである。 

④ 指定された行ラベルに制御が移った後は、通常のプログラムと同じ動作をするが、

RETURN文まで来ると、先に実行された LINK文の「次の命令文」に制御が自動的に戻される。 
 

このプログラムでは、まず変数 YEARに変数 birthyr を代入後、LINK文で行ラベル CONVERTに

制御を飛ばす。飛ばされた先で初めて出会う RETURN 文によってもと来た所に戻されて、次の命令

文を実行することになり、換算結果の値を持つ変数 YEAR を変数 birthyrに代入する。こうすること

で、何度でも開いたサブルーティンに飛んで、計算が終われば「自動的に」もと来た場所に戻って

くることが出来るので、前節のように煩わしいコピー･ペースト作業をする必要はない。同じことを変

数 wedyrにも施した後、引き算をして結婚した年齢 wedageが計算できる。つまり、変数 YEARは「仮

引数」の役割を果たしている。 

実は、ここで極めて重要なことがある。メインプログラムに相当する部分の最後に return 文があ

る。この return文と「開いたサブルーティン」部の最後にある RETURN文は共に「明示的 return文」

(explicit return)と呼ばれるが、役割は同じではない。「メインプログラムに相当」にある return文は、

これが実行されると

現在のオブザベー

ション (正確には、

PDV(プログラム・デ

ータベクトル)をデー

タセットに保存した

後、次のオブザベー

ションに処理が移る。

仮に、この return文

がないとすると、すぐ

下 の 行 ラ ベ ル

CONVERT に あ る

select文を実行され

てしまう。わかりやす

く言えば、この明示

的 return 文が関所

の役割を果たして、

その下の命令文には制御が行かないように阻止している。 

*★★★ ③LINK-RETURN 文使用のプログラム ★★★; 
data link_return (drop=startyear YEAR); set FCMP.wedage; 
  YEAR=birthyr; LINK CONVERT; birthyr=YEAR; 
  YEAR=wedyr;   LINK CONVERT; wedyr  =YEAR; 
  wedage=wedyr-birthyr; 
  return; *明示的 return文; 
 
  CONVERT: *行ラベル; 
  select(substr(YEAR,1,2)); *年を西暦変換; 
    when("明") startyear=1867;  
    when("大") startyear=1911;  
    when("昭") startyear=1925;  
    when("平") startyear=1988;  
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then YEAR=startyear+substr(YEAR,3,2); 
                   else YEAR=startyear+YEAR; 
  RETURN; 
run; 
 
proc print data=link_return; title "③LINK-RETURN 文使用"; run; 

メインプログラムに相当 

開いたサブルーティン 



237    %convert(birthyr); 
MACROGEN(CONVERT):   select(substr(birthyr,1,2)); 
MACROGEN(CONVERT):   *年を西暦変換; 
MACROGEN(CONVERT):   when("明") startyear=1867; 
MACROGEN(CONVERT):   when("大") startyear=1911; 
MACROGEN(CONVERT):   when("昭") startyear=1925; 
MACROGEN(CONVERT):   when("平") startyear=1988; 
MACROGEN(CONVERT):   when("19") startyear=0; 
MACROGEN(CONVERT):   when("20") startyear=0; 
MACROGEN(CONVERT):   otherwise startyear=.; 
MACROGEN(CONVERT):   *欠損値又はエラー; 
MACROGEN(CONVERT):   end; 
MACROGEN(CONVERT):   if startyear > 0 then 

 birthyr=startyear+substr(birthyr,3,2); 
MACROGEN(CONVERT):   else birthyr=startyear+birthyr; 
238    %convert(wedyr  ); 
MACROGEN(CONVERT):   select(substr(wedyr,1,2)); 
MACROGEN(CONVERT):   *年を西暦変換; 
MACROGEN(CONVERT):   when("明") startyear=1867; 
MACROGEN(CONVERT):   when("大") startyear=1911; 
MACROGEN(CONVERT):   when("昭") startyear=1925; 
MACROGEN(CONVERT):   when("平") startyear=1988; 
MACROGEN(CONVERT):   when("19") startyear=0; 
MACROGEN(CONVERT):   when("20") startyear=0; 
MACROGEN(CONVERT):   otherwise startyear=.; 
MACROGEN(CONVERT):   *欠損値又はエラー; 
MACROGEN(CONVERT):   end; 
MACROGEN(CONVERT):   if startyear > 0 then 
       wedyr=startyear+substr(wedyr,3,2); 
MACROGEN(CONVERT):   else wedyr=startyear+wedyr; 

LOG 画面の 

マクロ展開結果 

8. SAS マクロ使用のプログラム 
第４節で示した和暦を

西暦に換算するアルゴリズ

ムをSASのマクロ言語で表

現したコードを右に示す。

このマクロを実行するプロ

グラムはその下に示す。 

実際の和暦・西暦換算

は、data ステップにある  

%convert(birthyr);  

%convert(wedyr); 

で行われるが、SAS が実

際に実行するのは、このマ

クロを展開した結果、自動

作成されたSASコードであ

る。「options macrogen;」

を指定しておけばその展

開結果が以下のように、

LOG 画面に表示される。

この LOG 画面に表示されたプログラムは第５節で示した初心者レベルのプログラムそのものである。つ

まり、マクロで表現しておけば、実行時に(正確にはコンパイル時に)SAS システムが自動でコピー・ペー

スト作業をしてくれるので、マニュアル作業で行う際のうっかりミスはまったくなく安全である。ユーザーに

してみれば、convert というマクロを一度定義しておけば、あたかもサブルーティンのように使え、メイン

プログラムに相当する data ステップは極めて簡潔で読みやすくなる。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

*★★★ ④SAS マクロ使用のプログラム ★★★; 
%macro convert(YEAR); 
  select(substr(&YEAR,1,2)); *年を西暦変換; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then &YEAR=startyear+substr(&YEAR,3,2); 
                   else &YEAR=startyear+&YEAR; 
%mend convert; 

options macrogen; 
 
data macro_use (drop=startyear); set FCMP.wedage; 
  %convert(birthyr); 
  %convert(wedyr  ); 
  wedage=wedyr-birthyr; 
run; 
 
proc print data=macro_use; title "③SAS マクロ使用"; run; 



*★★★ ⑤PROC FCMP で function 作成 ★★★; 
libname sasFCMP "F:\sasFCMP\functions\conversion"; 
 
proc fcmp outlib=sasFCMP.functions.conversion; 
  FUNCTION nengo(YEAR $); ********************************; 
    select(substr(YEAR,1,2)); *生年を西暦変換; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then xyear=startyear+substr(YEAR,3,2); 
                     else xyear=startyear+YEAR; 
      return(xyear); 
  ENDSUB; *************************************************; 
run; 

9. PROC FCMP で関数の作成 
本節では SAS9.2 から追加された新機能の proc fcmp を使って和暦・西暦換算をする関数 nengo

を作成する。この関数を定義するコードを以下に示す。関数作成時の注意事項としては、最初に

libname 文で３層構造のフォルダを定義しておかなければならない。次に、proc fcmp 文の中で outlib=

オプションを使って、そのフォルダに対応する３次のライブラリ参照名を指定する。そうすることで最終的

に、この例では、「F:\sasFCMP\functions\conversion」のフォルダの中に二つの SAS ファイルが作成さ

れる。ひとつは functions.sas7bdatというSASデータセットが作成されており、ダブルクリックして開いてみ

ると、その中に「value」という名前の変数があり、この関数のSASコードがテキスト形式でそっくり保存され

ているのが分かる。もうひとつは functions.sas7bndxというSASファイルで、その中にはオブジェクトコード

化された関数が保存され、この関数の使用時にはこれが直接使われると思われる。 

関数を定義する本体

部分は、FUNCTION 文か

ら ENDSUB文までである。

FUNCTION文で関数名お

よび仮の引数とその変

数型を定義する。デフォ

ルトの変数型は数値型

なので、その場合には

変数の型を指定しなくて

済む。文字型なら「$」を

指定する。計算処理を

する部分は、data ステッ

プで使える命令文を使

って記述する。この関数

の計算結果を保存している変数 xyear を return 文で指定すると、この値が関数値となる。 

実際にこの関数を使う dataステップを以下に示す。ここで注意しておくことは、dataステップで使用す

る前に options文で cmplib=を使って、先に proc fcmp outlib=で指定した参照名の先頭の二つから

なる２次名を指定することである。これは以下に示すコードのコンパイル時に必要な関数の保存場所で

あり、この例では、フォルダ sasFCMPの中にあるフォルダ functionsの中に当該関数があることを示して

いる。data ステップで使用する際は、通常の関数を使用するのと同じようにすればよい。 

 

 

 

 

 

10.PROC FCMP でサブルーティンの作成 
和暦・西暦換算をするサブルーティン nengo_conv を proc fcmp で作成するのは、先の関数を作成

するのとほぼ同じ手順でできる。異なる点は三点有り、①SUBROUTINE 文でサブルーティン名を定義して、

かつ、必要なら、メインプログラムに相当する data ステップへ引き渡す仮引数と、data ステップに戻す

仮引数を指定すること、②outargs文で dataステップに戻す仮引数を定義しておくこと、③計算結果は

実引数に持たせるので、関数のような return命令は要らないことである。 

options cmplib=(sasFCMP.functions); 
 
data function_use; set FCMP.wedage; 
  wedage=nengo(wedyr)-nengo(birthyr); 
run; 
 
proc print data=function_use; title "⑤PROC FCMP で function 作成"; run; 



*★★★ ⑥PROC FCMP で subroutine 作成 ★★★; 
libname sasFCMP "G:\sasFCMP\functions\conversion"; 
 
proc fcmp outlib=sasFCMP.functions.conversion; 
  SUBROUTINE nengo_conv(YEAR $,xyear); ********************; 
    outargs xyear; *メインプログラムに引き渡す仮引数; 
    select(substr(YEAR,1,2)); *生年を西暦変換; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then xyear=startyear+substr(YEAR,3,2); 
                     else xyear=startyear+YEAR; 
  ENDSUB; *************************************************; 
run; 

options cmplib=(sasFCMP.functions); 
 
data subroutine_use (drop=xbirthyr xwedyr); set FCMP.wedage; 
  xbirthyr=.; call nengo_conv(birthyr,xbirthyr); 
  xwedyr  =.; call nengo_conv(wedyr  ,xwedyr  ); 
  wedage=xwedyr-xbirthyr; 
run; 

このサブルーティンをdataステップで使用するには、普通にcall文を使い、必要に応じて実引数を

指定する。以下の例では、生年の変数(birthyr)と結婚した年の変数(wedyr)からそれぞれ西暦換算し

た新しい変数 xbirthyr と xwedyrが作成される。ここで気になるのは、この二つの新変数に事前に欠損

値を与えて初期化していることである。実はこれがなくても変数 wedage は正しく計算されるが、LOG 画

面には、「NOTE: 変数 xbirthyr は初期化されていません。NOTE: 変数 xwedyr は初期化されてい

ません。」というメッセージが表示される。確かに、｢xbirthyr=.;｣がなければ、変数 xbirthyrは初めて

現れた変数なので初期化されていないのは当たり前なのだが、サブルーティン実行後には値が正しく

与えられるから問題ないはずである。エラーメッセージでも警告メッセージでもないので、放っておいて

問題はないと判断するが、現段階ではとりあえずすべての実引数に欠損値を call 直前で与えておくこ

とにするが、いずれ将来この LOG にある余計なメッセージは出なくなると思う。 

 

 

 

 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 

 
 

11.まとめ 
SAS では、これまで自前の関数とサブルーティンを作成する機能がなく、それに代わる方法をいろい

ろ工夫しながら考えて切り抜けてきた。それらの技法をまず解説した後、SAS9.2 から追加された新機

能の proc fcmpを使って、和暦・西暦換算をする関数 nengo とサブルーティン nengo_convを作成し、

自前の関数とサブルーティンの作り方と使用法を解説した。 

そもそも proc fcmpが出現したことを当初は知らなかった。2012年に米国フロリダ州で開催された

SAS の世界大会(SAS Global Forum 2012)で数独を解く SAS プログラムを招待論文として筆者が報

告した際、その準備段階でカナダ人の世話人(参考文献１)から、自分たちは proc fcmpを使って数

独プログラムを作る旨のメールを受け取り、その存在を知った次第である。 



本稿で示したように、自前の関数とサブルーティンが手軽に作れるようになると、data ステップが格

段に簡潔になり、見やすく、しかも信頼度の高いプログラムになる。今後ユーザーの皆さんにもぜひ活

用されることを大いに望んでいる。参考文献２は分かりやすく proc fcmpの解説をしている。 

付録 SAS プログラム 
 

/* 婚姻年齢.sas */ options nocenter; 
libname FCMP "G:\sasFCMP\sasds"; 
 
proc import out=FCMP.wedage 
    datafile="G:\sasFCMP\婚姻年齢.xlsx" 
    dbms=excel replace; 
    getnames=yes; 
    sheet="sheet1"; 
run; 
 
proc print; title "data=wedage"; run; 
 
*★★★ ①初心者のプログラム ★★★; 
data beginner (drop=startyear); set FCMP.wedage; 
  select(substr(birthyr,1,2)); *★★生年を西暦変換★★; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then birthyr=startyear+substr(birthyr,3,2); 
                   else birthyr=startyear+birthyr; 
 
  select(substr(wedyr,1,2));  *★★結婚した年を西暦変換★★; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then wedyr=startyear+substr(wedyr,3,2); 
                   else wedyr=startyear+wedyr; 
 
  wedage=wedyr-birthyr; *結婚した時の年齢計算; 
run; 
 
proc print data=beginner; title "①初心者のプログラム"; run; 
 



*★★★ ②ARRAY 文使用のプログラム ★★★; 
data array_used (drop=i startyear); set FCMP.wedage; 
  array arg $ birthyr wedyr; 
  do i=1 to 2; 
    select(substr(arg{i},1,2)); *★★年を西暦変換★★; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then arg{i}=startyear+substr(arg{i},3,2); 
                     else arg{i}=startyear+arg{i}; 
  end; 
  wedage=wedyr-birthyr; *結婚した時の年齢計算; 
run; 
 
proc print data=array_used; title "②ARRAY 文使用"; run; 
 
*★★★ ③LINK-RETURN 文使用のプログラム ★★★; 
data link_return (drop=startyear YEAR); set FCMP.wedage; 
  YEAR=birthyr; LINK CONVERT; birthyr=YEAR; 
  YEAR=wedyr;   LINK CONVERT; wedyr  =YEAR; 
  wedage=wedyr-birthyr; 
  return; 
 
  CONVERT: *行ラベル; 
  select(substr(YEAR,1,2)); *年を西暦変換; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then YEAR=startyear+substr(YEAR,3,2); 
                   else YEAR=startyear+YEAR; 
  RETURN; 
run; 
 
proc print data=link_return; title "③LINK-RETURN 文使用"; run; 
 
*★★★ ④SAS マクロ使用のプログラム ★★★; 
%macro convert(YEAR); 
  select(substr(&YEAR,1,2)); *年を西暦変換; 
    when("明") startyear=1867; 
    when("大") startyear=1911; 
    when("昭") startyear=1925; 
    when("平") startyear=1988; 
    when("19") startyear=0; 
    when("20") startyear=0; 
    otherwise  startyear=.; *欠損値又はエラー; 
  end; 
  if startyear > 0 then &YEAR=startyear+substr(&YEAR,3,2); 
                   else &YEAR=startyear+&YEAR; 
%mend convert; 
 
options macrogen; 
 
data macro_use (drop=startyear); set FCMP.wedage; 
  %convert(birthyr); 
  %convert(wedyr  ); 
  wedage=wedyr-birthyr; 
run; 



proc print data=macro_use; title "④SAS マクロ使用"; run; 
 
*★★★ ⑤PROC FCMP で function 作成 ★★★; 
libname sasFCMP "G:\sasFCMP\functions\conversion"; 
 
proc fcmp outlib=sasFCMP.functions.conversion; 
  FUNCTION nengo(YEAR $); ********************************; 
    select(substr(YEAR,1,2)); *年を西暦変換; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then xyear=startyear+substr(YEAR,3,2); 
                     else xyear=startyear+YEAR; 
      return(xyear); 
  ENDSUB; *************************************************; 
run; 
 
options cmplib=(sasFCMP.functions); 
 
data function_use; set FCMP.wedage; 
  wedage=nengo(wedyr)-nengo(birthyr); 
run; 
 
proc print data=function_use; title "⑤PROC FCMP で function 作成"; run; 
 
*★★★ ⑥PROC FCMP で subroutine 作成 ★★★; 
libname sasFCMP "G:\sasFCMP\functions\conversion"; 
 
proc fcmp outlib=sasFCMP.functions.conversion; 
  SUBROUTINE nengo_conv(YEAR $,xyear); ********************; 
    outargs xyear; *メインプログラムに引き渡す仮引数; 
    select(substr(YEAR,1,2)); *年を西暦変換; 
      when("明") startyear=1867; 
      when("大") startyear=1911; 
      when("昭") startyear=1925; 
      when("平") startyear=1988; 
      when("19") startyear=0; 
      when("20") startyear=0; 
      otherwise  startyear=.; *欠損値又はエラー; 
    end; 
    if startyear > 0 then xyear=startyear+substr(YEAR,3,2); 
                     else xyear=startyear+YEAR; 
  ENDSUB; *************************************************; 
run; 
 
options cmplib=(sasFCMP.functions); 
 
data subroutine_use (drop=xbirthyr xwedyr); set FCMP.wedage; 
  xbirthyr=.; call nengo_conv(birthyr,xbirthyr); 
  xwedyr  =.; call nengo_conv(wedyr  ,xwedyr  ); 
  wedage=xwedyr-xbirthyr; 
run; 
 
proc print data=subroutine_use; title "⑥PROC FCMP で subroutine 作成"; run; 


